query: 
  SELECT
    # Number of works on the topic with the author name string
    ?count
  
    # Number of works with the author name string
    ?count1
  
    # Author as a string
    ?authorstring
  
    # Build URL to the Author disambiguator tool
    (CONCAT(
        'https://tools.wmflabs.org/author-disambiguator/names_oauth.php?name=',
        ENCODE_FOR_URI(?authorstring)) AS ?resolver_url)
  WITH {
    SELECT DISTINCT ?work ?authorstring WHERE {
      ?work wdt:P2093 ?authorstring .
    ?work wdt:P31 wd:Q13442814 .              # scholarly articles
  
    { ?work wdt:P921 wd:Q84263196 . }         # with main subject "COVID-19"
    UNION                                  # or
    { ?work wdt:P921 wd:Q82069695 . }         # with main subject "SARS-CoV-2"
                                           # or
    { ?work wdt:P50 [?p1 wd:Q668] . }         # with an author linked to India
    UNION                                  # or
    { ?work wdt:P50 [?p2 wd:Q6973748] . }     # with an author linked to the National Institute of Virology
    UNION                                  # or
    { ?work wdt:P50 [wdt:P166 wd:Q372007] . } # with an author having received the Shanti Swarup Bhatnagar Prize for Science and Technology
    }
    LIMIT 1000
  } AS %works
  WITH {
    SELECT
      (COUNT(DISTINCT ?work) AS ?count)
      (COUNT(DISTINCT ?work1) AS ?count1)
      ?authorstring
    WHERE {
      INCLUDE %works
      ?work1 wdt:P2093 ?authorstring .
    }
    GROUP BY ?authorstring
  } AS %result
  WHERE {
    INCLUDE %result
  }
  ORDER BY DESC(?count) DESC(?count1)
  #LIMIT 50
  
    
  